Interpolación polinómica en una variable

Problemas de interpolación polinómica más habituales

La orden **InterpolatingPolynomial** permite obtener el polinomio de interpolación cuyos datos son lagrangianos, Hermite, Taylor o Hermite generalizado. Hay que dar los datos entre llaves y a continuación, sepada por coma, la variable del polinomio interpolador. Si en un nodo se tienen como datos el valor de la función y el de 5 derivadas consecutivas, por ejemplo, los 6 datos van en una llave que sigue al nodo. Veamos un ejemplo; vamos a calcular el polinomio cuyos datos en el punto 1 son {5,1,0}, es decir, vale 5, su derivada primera vale 1 y su derivada segunda vale 0; los datos en el punto 3 son {1,3} y, finalmente, su valor en 5 es -2. El polinomio lo damos en la variable t. Observe la forma diferente de escribir los datos, entre llaves, cuando sólo hay un dato, el valor de la función, y cuando hay más de un dato.

```
datos = {{1, {5, 1, 0}}, {3, {1, 3}}, {5, -2}};
pol[t_] = Expand[InterpolatingPolynomial[datos, t]]
```

Vamos a emplear esta orden para resolver directamente varios problemas clásicos de interpolación polinómica.

■ Interpolación polinómica lagrangiana

Calculemos el polinomio que pasa por los puntos (1,1), (2,0), (4,0), (5,1) y (6,1.5). Para ello, definamos los datos y utilicemos la orden **InterpolatingPolynomial**. Para ver claramente el resultado utilizamos la orden **Expand**.

```
datos1 = {{1, 1}, {2, 0}, {4, 0}, {5, 1}, {6, 1.5}};
pol1[t_] = Expand[InterpolatingPolynomial[datos1, t]]
```

Ahora definimos la representación gráfica de los puntos interpolados, para lo que utilizamos la orden **ListPlot**. Para que no se muestre de forma automática empleamos la opción **DisplayFunction→Identity**. Actuamos de igual modo para definir la gráfica del polinomio de interpolación.

```
g1 = ListPlot[datos1, PlotStyle → {PointSize[0.03]}, DisplayFunction → Identity];
g2 = Plot[pol1[t], {t, 1, 6}, DisplayFunction → Identity];
```

Mostramos conjuntamente las gráficas definidas, indicando que el resultado sea mostrado por pantalla, lo que se logra empleando la opción **DisplayFunction→$DisplayFunction**.

```
Show[{g1, g2}, DisplayFunction → $DisplayFunction];
```

Calculemos ahora el polinomio de interpolación de la función $\frac{\mathrm{Exp}[x]}{6x^2+5}$ en los puntos obtenidos al dividir el intervalo [-1,3] en 4 partes iguales; es decir, determinemos el polinomio que interpola los valores de la función $\frac{\mathrm{Exp}[x]}{6x^2+5}$ en los puntos -1, 0 ,1, 2 y 3. Seguimos el procedimiento del ejemplo anterior, pero comenzamos definiendo los puntos que deben ser interpolados.

```
datos2 = Table[{x, E^x/(6 x^2 + 5)}, {x, -1., 3}];
pol2[t_] = Expand[InterpolatingPolynomial[datos2, t]]

g3 = ListPlot[datos2, PlotStyle → {PointSize[0.02]}, DisplayFunction → Identity];
g4 = Plot[pol2[t], {t, -3, 4},
   DisplayFunction → Identity, PlotStyle → {Dashing[{0.01, 0.02}]}];
g5 = Plot[E^x/(6 x^2 + 5), {x, -3., 4}, DisplayFunction → Identity];

Show[{g3, g4, g5}, DisplayFunction → $DisplayFunction];
```

■ Interpolación polinómica de Hermite clásica

Hallemos ahora el polinomio que pasa por los puntos (1,1), (2,0), (4,0), (5,1) y (6,1.5) siendo su derivada primera nula en todos ellos:

```
datos3 = {{1, {1, 0}}, {2, {0, 0}}, {4, {0, 0}},
   {5, {1, 0}}, {6, {1.5, 0}}};
pol3[t_] = Expand[InterpolatingPolynomial[datos3, t]]

g6 = Plot[pol3[t], {t, 1, 6.1}, DisplayFunction → Identity];
Show[{g6, g1}, DisplayFunction → $DisplayFunction];
```

Calculemos el polinomio que interpola los valores de la función $\frac{\mathrm{Exp}[x]}{6x^2+5}$ y los de su derivada en los puntos -1, 0, 1, 2 y 3:

```
datos4 = Table[{x, {E^x/(6 x^2 + 5), ∂{z,1} E^z/(6 z^2 + 5) /. z → x}}, {x, -1., 3}]
pol4[t_] = Expand[InterpolatingPolynomial[datos4, t]]

g7 = Plot[pol4[t], {t, -2, 4},
   PlotStyle → {Dashing[{0.025, 0.025}]}, DisplayFunction → Identity];

Show[{g3, g5, g7}, DisplayFunction → $DisplayFunction];
```

Se observa una mayor coincidencia de este interpolante y la función interpolada en el intervalo $[-1, 3]$ que del interpolante de Lagrange en los mismos puntos, visto más arriba. Sin embargo, fuera de dicho intervalo la diferencia es mayor en este caso.

■ Interpolación polinómica de Taylor

Finalmente vamos a calcular el interpolante de Taylor, de grado 5, para la misma función que venimos considerando, y centrado en el punto $x = 1$ (nodo central de los considerados anteriormente).

```
datos5 = {{1, Table[∂{z,i} E^z/(6 z^2 + 5) /. z → 1., {i, 0, 4}]}}
pol5[t_] = Expand[InterpolatingPolynomial[datos5, t]]

g8 = Plot[pol5[t], {t, -1, 3.},
   PlotStyle → {Dashing[{0.025, 0.025}]}, DisplayFunction → Identity];

Show[{g5, g8}, DisplayFunction → $DisplayFunction];
```

■ Fórmula de Lagrange para la interpolación polinómica lagrangiana

Vamos a utilizar la fórmula de Lagrange para construir el polinomio de interpolación de Lagrange del problema en que los nodos de interpolación son 0, 1, 2, ..., 10 (guardados en una lista *x*) y los valores a interpolar son números aleatorios comprendidos entre 0 y 8 (que guardamos en un vector y). Se generan a partir de la orden **Random**[].

```
Clear["Global`*"]
```

```
n = 12;
x = Table[i, {i, 0, n}];
y = Table[8 Random[], {i, 0, n}]
```

El polinomio de interpolación de Lagrange se puede formar directamente:

$$\texttt{pLagrange[t\_] :=} \sum_{i=1}^{n+1} y[\![i]\!] \left(\prod_{j=1}^{i-1} \frac{t - x[\![j]\!]}{x[\![i]\!] - x[\![j]\!]} \prod_{j=i+1}^{n+1} \frac{t - x[\![j]\!]}{x[\![i]\!] - x[\![j]\!]} \right);$$

Para comprobar que la interpolación ha sido correcta podemos evaluar la función precedente en la lista de nodos, **x**, debiendo obtener el vector de valores interpolados, **y**.

```
pLagrange[x]
```

Podemos representar la gráfica del polinomio de interpolación que hemos calculado y medir el tiempo que se necesita para obtenerla, que dependerá, lógicamente, de las prestaciones del equipo que estemos empleando. La orden correspondiente es **Timing**, que produce una lista de salida con dos posiciones. Tenga paciencia.

```
Timing[Plot[pLagrange[t], {t, x[[1]], x[[n + 1]]}]]
```

El núcleo de *Mathematica* ha empleado un tiempo elevado para producir la gráfica.

Ejercicios

1.- Se considera la función no negativa $\text{Exp}[-(3.9 + \frac{5}{2000})\, t]\, t^2$. Calcule su polinomio de interpolación en los nodos 0,1,2,3,4,5. Represéntelo gráficamente y calcule su valor en el punto 4.75.

2.- Interpole la función $\text{Exp}[x] + \frac{x}{5}$ en los puntos -n,...,-1,0,1,...,n para n=1, 2, 3. Represente la función y los interpolantes en el intervalo [-3,3] utilizando diferentes colores.

3.- Para la misma función del ejercicio anterior, halle el interpolante de Hermite en el espacio de los polinomios de grado menor o igual que trece para los nodos -3,-2,-1,0,1,2 y 3. Represéntelo en [-3,3].